```
"Define functions for basic calculations"
FUNCTION a(n)
{?AVALUE
    This function calculates the value of a}

if (n<0) then
    a:=0.15  "a is the coefficient in the equation add number"
    else
    a:=0.3
    endif
END

FUNCTION T_snow(T_s)
  {$TSNOW
    This function return the the snow surface temperature from the equation }
    if (T_s>0) then
    T_snow:=0
    else
    T_snow:=T_s
    endif
END


"Equations for unit conversion"
T_a=(T_ori-32)*5/9 "temperature has unit of C"
U=U_ori*convert(mph,m/s) "convert unit"
RH_a=RH_ori*0.01
Q_si=Q_ori_sn*convert(W/m^2,KJ/(m^2*s)) "convert unit"
Q_g=Q_ori_g*convert(btu/(hr*ft^2),KJ/(m^2*s)) "convert unit"
p_sat_a=pressure(steam, T=T_a, x=1) "calculate steam pressure"

e_a=RH_a*p_sat_a*10
"e_a is mbar"

"heat flux for long wave radiation"
Q_li_1=(sigma*(T_a+273)^4)*1.08*(1-exp(-e_a^((T_a+273)/2.016)))*convert(W/m^2,KJ/(m^2*s))
sigma=5.6698*10^(-8)
Q_li=Q_longwave*convert(W/m^2,KJ/(m^2*s))*epsilon

"energy equation related to snow melting, this equation is used to calculate surface temperature."
Q=Q_sn+Q_li-Q_le+Q_h+Q_e-Q_g "equation 10 in the manuscript, Q is the energy for snow melting, and
has unit of KJ/(m^2*s)"
Q=h*rho_snow*333.5*B/(9*3600) "equation 11, unit:kj/m2/s"
B=1-V_pore
Q_sn=(1-Albedo)*Q_si  "short wave radiation: albedo obtained from the table in reference #"
time=row

"constants"
epsilon=0.97
D_H=0.00574
D_E=0.01
Q_le_1=epsilon*sigma*((T_s+273)^4)/(10^3) "equation 14"
Q_le=Q_le_1
Q_h=D_H*U*(T_a-T_s) "equation 15"
e_s=pressure(steam, T=T_s, x=1)*10  "unit:mbar"
Q_e=D_E*U*(e_a-e_s) "equation 16"
```

```
rho_ice=0.9167 "g/cm^3"
K_ice=2.2 "w/mk"
K_water=0.5475
K_air=Conductivity(Air,T=T_snowpack)

"calculate snow temperatures"
T_snow=T_snow(T_s)
T_snowpack_0=(T_snow+T_r)/2    "equation 6 in the manuscript"
T_snowpack=T_snow(T_snowpack_0)
a=a(T_snowpack)
delta_T=T_snow-T_r    "temperatures in Celcius"

"energy balance equation for snow layer, this equation is used to calculate heat flux for thermal conduction"
Q_snow=Q_g+Q  "equation 17, Q_cond refers to heat flux for thermal conduction."

"Calculate snow conductivity K_snow with calculated surface temperature and heat flux of thermal conduction"
K_snow_0=Q_snow*1000*l/delta_T "Q_snow:kW/m^2, l:m, K:W/mK"  "equation 3 in the manuscript"
K_snow=abs(K_snow_0)

"Calculate K_sat and K_dry, equations from Johansen method"
V_pore=1-rho_snow*0.001/rho_ice            "rho:kg/m^3, 920 is ice density"
V_air=V_pore-V_water
K_sat=K_ice^(1-V_pore)*K_water^V_pore
K_dry=((K_ice*a-K_air)*rho_snow*0.001+K_air*rho_ice)/(rho_ice-(1-a)*rho_snow*0.001) "equation 5 in the manuscript"

"Calculate Ke number"
K_e=(K_snow-K_dry)/(K_sat-K_dry)  "equation 1 in the manuscript"

"Calculate theta"
theta=V_water/V_pore
log_theta=log10(theta)
```